/*

Adam Howat	
Proposal title: Not all stereotypes are equal: Consequences of partisan stereotypes on polarization



HYPOTHESES

Stated-Hyp1: Trait-based thinking about partisanship will lead to greater affective 
	polarization and perceived ideological distance between the parties, as 
	compared to group- or issue-based thinking.

	Test-Hyp1: Trait-based thinking about partisanship will lead to greater affective 
	polarization between the parties, as compared to group-based thinking.
	
	Test-Hyp2: Trait-based thinking about partisanship will lead to greater perceived ideological distance between the parties, as compared to group-based thinking.

	Test-Hyp3: Trait-based thinking about partisanship will lead to greater affective 
	polarization between the parties, as compared to issue-based thinking.
	
	Test-Hyp4: Trait-based thinking about partisanship will lead to greater perceived ideological distance between the parties, as compared to issue-based thinking.
	
Stated-Hyp2: Trait-based thinking about partisanship will lead to stronger partisan 
	social identity, as compared to group- or issue-based thinking.

	Test-Hyp5: Trait-based thinking about partisanship will lead to stronger partisan 
	social identity, as compared to group-based thinking.

	Test-Hyp6: Trait-based thinking about partisanship will lead to stronger partisan 
	social identity, as compared to issue-based thinking.

*/

use "Howat1039.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION

* variable identifying treatment groups
	clonevar treat = P_EXP29
	tab treat
	
* OUTCOME

* affective polarization
	/* absolute difference in feeling thermometer for dems and republicans */
	
	* recode missing data
	replace Q2_D=. if Q2_D>100
	replace Q2_R=. if Q2_R>100
	
	gen pol_affect= abs(Q2_D - Q2_R)
	tab pol_affect

* percieved ideological distance b/w parties
	/* Respondents were asked: Where would you place dems and reps on 
	a scale of liberal to conservative.
	The distance between the two is percieved ideological distance
	*/
	
	// recode missing
	foreach var of varlist Q3* {
	replace `var'=. if `var'>7
	tab `var', mis
	}
	
	** distance b/w supporters of the two parties	
	gen pol_partymember = abs(Q3_1	- Q3_3)
	tab pol_partymember, mis
	
	** members of congress
	gen pol_congress = abs(Q3_2-Q3_4)
	tab pol_congress, mis

	* creating a summated measure of perception of political polarization
	gen pol_perception = pol_partymember + pol_congress 	
	tab pol_perception
	
* partisan social identity
	/* importance of party id */
	tab Q4
	clonevar partisan_id= Q4
	replace partisan_id=. if partisan_id>5
	tab partisan_id, mis
	
********************************************************************************

* ANALYSIS
	
*Test-Hyp1: Trait-based thinking about partisanship will lead to greater affective polarization between the parties, as compared to group-based thinking.
	
	reg pol_affect ib3.treat if treat==1|treat==3
	// reject. 0.694
	tess 1.treat +, init(Howat1039) bonf(3)
	
*Test-Hyp2: Trait-based thinking about partisanship will lead to greater perceived ideological distance between the parties, as compared to group-based thinking.

	reg pol_perception ib3.treat if treat==1|treat==3
	// reject. 0.952
	tess 1.treat +, bonf(3)	

*Test-Hyp3: Trait-based thinking about partisanship will lead to greater affective polarization between the parties, as compared to issue-based thinking.
	
	reg pol_affect ib2.treat if treat==1|treat==2
	// reject. 0.905
	tess 1.treat +, bonf(3)
	
*Test-Hyp4: Trait-based thinking about partisanship will lead to greater perceived ideological distance between the parties, as compared to issue-based thinking.
	
	reg pol_perception ib2.treat if treat==1|treat==2
	// reject. 0.018
	* bonferroni correction: test threshold is 0.016 b/c there are 3 outcomes per compared treatment pairs
	tess 1.treat +, bonf(3)
	
*Test-Hyp5: Trait-based thinking about partisanship will lead to stronger partisan social identity, as compared to group-based thinking.

	reg partisan_id ib3.treat if treat==1|treat==3
	// reject. 0.234
	tess 1.treat +, bonf(3)
	
*Test-Hyp6: Trait-based thinking about partisanship will lead to stronger partisan social identity, as compared to issue-based thinking.

	reg partisan_id ib2.treat if treat==1|treat==2
	// reject. 0.304
	tess 1.treat +, bonf(3)
